


総合演習I

担当: 佐藤

計算機実習III

第4回: 演習課題解説






課題4-1

解答例

```
size(500, 500) ;  
strokeWeight(2) ;  
translate(width / 2, height / 2) ;  
for (int length = 10 ; length < width ; length += 10) {  
  line(0, 0, length, 0) ;  
  translate(length, 0) ;  
  rotate(HALF_PI) ;  
  line(0, 0, length, 0) ;  
  translate(length, 0) ;  
  rotate(HALF_PI) ;  
}
```

「┐」を1セットとして描くことを繰り返す




課題4-2

解答例

```
void setup() {  
  size(300, 300) ;  
  background(255) ;  
  noFill() ;  
  rectMode(CENTER) ;  
  strokeWeight(2) ;  
  for (int x = 0 ; x < width + 40 ; x += 80) {  
    for (int y = 0 ; y < height + 70 ; y += 140) {  
      hexBar(x, y) ;  
      hexBar(x + 40, y + 70) ;  
    }  
  }  
}
```

```
void hexBar(float x, float y) {  
  pushMatrix() ;  
  translate(x, y) ;  
  for (int i = 0 ; i < 3 ; i++) {  
    rotate(PI / 3) ;  
    rect(0, 0, 30, 100) ;  
  }  
  popMatrix() ;  
}
```



課題4-3

解答例

```
float diameter = 180 ;  
float angle = 0 ;  
boolean isStop = false ;
```

```
void setup() {  
  size(400, 400) ;  
  noFill() ;  
  strokeWeight(3) ;  
  colorMode(HSB) ;  
}
```


```
void mousePressed() {  
  if (isStop == false)  
    noLoop() ;  
  else loop() ;  
  isStop = !isStop ;  
}
```

```
void draw() {  
  background(255) ;  
  translate(width / 2, height / 2) ;  
  rotate(angle) ;  
  for (int i = 0 ; i < 360 ; i++) {  
    float theta = radians(i) ;  
    float x = diameter / 2 * cos(theta) ;  
    float y = diameter / 2 * sin(theta) ;  
    stroke(float(i) / 360 * 255, 255, 255) ;  
    arc(x, y, diameter, diameter, theta, theta + PI, OPEN) ;  
  }  
  angle = (angle + 0.2) % TWO_PI ;  
}
```

半円の輪郭線

0.2[rad/frame]
回転(周期 2π)

360° で色が一巡
するよう角度と色
を対応させる



課題4-4

解答例

```
PVector[] vertices = new PVector[8] ;
```

```
void setup() {  
  size(500, 500) ;  
  translate(width / 2, height / 2) ;  
  for (int i = 0 ; i < vertices.length ; i++) {  
    PVector vertex = PVector.fromAngle(radians(0)) ;  
    vertex.mult(200) ;  
    if (i % 2 != 0) vertex.div(sqrt(2)) ;  
    vertex.rotate(i * QUARTER_PI) ;  
    vertices[i] = vertex ;  
  }  
  noStroke() ;  
  noLoop() ;  
}
```

奇数回のベクトルの
長さは「偶数回のベ
クトルの長さ/√2」

```
void draw() {  
  background(255) ;  
  translate(width / 2, height / 2) ;  
  drawWindmill() ;  
}
```

単に「0」でもよい

```
void drawWindmill() {  
  beginShape(TRIANGLE_FAN) ;  
  vertex(0, 0) ; // the center point  
  for (int i = 0 ; i < vertices.length ; i++) {  
    PVector v = vertices[i] ;  
    if (i % 2 != 0) fill(0) ;  
    else noFill() ;  
    vertex(v.x, v.y) ;  
  }  
  endShape(CLOSE) ;  
}
```

奇数回のみ図形を描く