

静止画の作成1

担当: 佐藤

計算機実習III

第1回



sato@ise.aoyama.ac.jp

2019/4/9



講義概要

- 使用言語

- ▶ Processing(プロセッシング)
- ▶ Processingの公式Webサイト(<https://processing.org/download/>)からダウンロード可能
- ▶ 講義室のPCにはあらかじめインストール済み

- 講義の構成

- ▶ 3部構成(1部あたり5回)
- ▶ 各部の最後の講義回→その部で学習した内容の「総合演習」

- 講義の流れ(総合演習回を除く)

- ▶ 前回課題の解説
- ▶ 学習内容の説明
- ▶ 課題に取り組む
- 終了時刻までに全課題を終えた者は退室可

- 教員

- ▶ 小林先生(コンビナー)
- ▶ 佐藤(第1部: 1～5回), 高橋先生(第2部: 6～10回), 杉之内先生(第3部: 11～15回)



講義計画

変更もあり得る

2D

- Processingの基礎
- 静止画の作成
- 動画の作成

3D

- ファイル操作と文字列処理
- イベントドリブン
- カメラ座標系

OOP

- オブジェクト指向プログラミング
- 物理シミュレーション
- フラクタル

開発環境



sato@ise.aoyama.ac.jp

2019/4/9




PDE

- Processing開発環境(Processing Development Environment: **PDE**)

ツールバー

メッセージエリア



メニュー


テキストエディタ

コンソール



スケッチ

- Processingで作成されたプログラムを**スケッチ**という
- スケッチの作成手順
 - ① コーディング
 - テキストエディタにコードを書く
 - ② 実行⇔デバッグ
 - ツールバーの「▶」ボタンを押すと実行
 - ツールバーの「■」ボタンを押すと停止
 - ③ 保存
 - メニューの「ファイル」→「名前を付けて保存」
 - そのスケッチ専用のフォルダ(**スケッチフォルダ**)が作成される
- スケッチを実行すると**ディスプレイウィンドウ**と呼ばれる新規ウィンドウが開く





スケッチの管理

【重要】

- **本講義で作成するスケッチの保存場所**

- ▶ **Z:¥My Documents¥Processing¥computerProgramming3¥xyyyyyyyy_zz¥**

- ① ZドライブのMy Documentsフォルダの直下に「Processing」フォルダを作成
- ② 「Processing」フォルダの直下に「computerProgramming3」フォルダを作成
- ③ 「computerProgramming3」フォルダの直下に「xyyyyyyyy_zz」フォルダを作成

- **xyyyyyyyy_zz**

- ▶ **x:** 自分の青山メールの最初の文字
- ▶ **yyyyyyy:** 自分の青山メールの2文字目以降の数字の並び
- (例)1234567
- ▶ **zz:** 講義回
- (例)第1回→「01」

- **「xyyyyyyyy_zz」フォルダの直下に毎回の講義で作成する各スケッチを保存**

- **保存したスケッチの開き方**

- ▶ スケッチフォルダの直下にある拡張子が「.pde」のファイルを開く



小演習

- **本講義で作成するスケッチフォルダの保存場所を全15回分作成しなさい**
 - ▶ 完成したら教員またはTAに確認してもらうこと



コード補完

- 準備

- ▶ メニューの「ファイル」→「設定」で開かれるウィンドウの中の「コードの補完」にチェックを入れると使用可能になる

- (例)関数のコード補完

- ▶ Processingが用意する関数を使う際，1文字以上打ち込んだ後に**Ctrl+スペース**で関数の候補リストが表示される
 - 候補リストの表示を消すにはEscキーを押す
- ▶ 上下の矢印キーで使いたい関数を選択するとコード補完が行われる



リファレンス

- Processingには公式のリファレンスが存在し，Webページとして公開されている
 - ▶ リファレンスには，Processingで用いられる様々な機能について個別に詳細な説明や使い方が書かれている
- PDEからリファレンスへアクセス可能
 - ▶ メニューの「ヘルプ」→「リファレンス」



座標，線，基本的な図形





座標

- コンピュータの画面は**ピクセル**と呼ばれる発光体が格子状に並んだものであり、各ピクセルは**座標**で示すことができる
- ピクセルの座標を(x, y)と表す
 - ▶ x: ディスプレイウィンドウの左端からの距離(ピクセル数)
 - ▶ y: ディスプレイウィンドウの上端からの距離(ピクセル数)
- (例)ディスプレイウィンドウのサイズが100×100ピクセル
 - ▶ 左上隅の座標: (0, 0)
 - ▶ 右下隅の座標: (99, 99)





size(), point()


- スケッチの構成単位は関数であり，あらかじめ用意された関数を使って基本的な図形を描くことができる

- **size()**: ディスプレイウィンドウの幅と高さを設定する関数

- ▶ 第1引数: 幅
- ▶ 第2引数: 高さ

例

```
size(80, 60);
```




size()が指定されていない場合，ディスプレイウィンドウのサイズは100×100ピクセルになる

- **point()**: ディスプレイウィンドウ内に点を描く関数

- ▶ 第1引数: x座標
- ▶ 第2引数: y座標

例

```
size(80, 60);  
point(40, 30);
```



原則として設定サイズとディスプレイウィンドウのサイズは一致する。しかし，ディスプレイウィンドウの最小サイズはOS依存であるため，最小サイズより小さいサイズを設定した場合，描画領域は最小サイズのディスプレイウィンドウの一部になる




line()

- **line():** 2点を結ぶ線を描く関数

- ▶ 第1引数: 点のx座標, 第2引数: 点のy座標
- ▶ 第3引数: 点のx座標, 第4引数: 点のy座標

例

```
size(200, 200);  
line(50, 150, 150, 50);
```






triangle(), quad()

- **triangle()**: 3点を結ぶ三角形を描く関数
 - ▶ 6つの引数(3つの点)
- **quad()**: 4点を結ぶ四角形を描く関数
 - ▶ 8つの引数(4つの点)


例

```
size(200, 200);  
triangle(100, 50, 50, 150, 150, 100);
```



例

```
size(200, 200);  
quad(100, 50, 50, 150, 100, 150, 150, 50);
```





rect(), ellipse()

- **rect():** 長方形を描く関数


- ▶ 第1引数: 左上の角の点のx座標, 第2引数: 左上の角の点のy座標
- ▶ 第3引数: 幅
- ▶ 第4引数: 高さ

- **ellipse():** 楕円を描く関数

- ▶ 第1引数: 中心のx座標, 第2引数: 中心のy座標
- ▶ 第3引数: 幅
- ▶ 第4引数: 高さ


例

```
size(200, 150);  
rect(50, 50, 100, 50);
```



例

```
size(200, 150);  
ellipse(100, 75, 100, 50);
```





rectMode()

- **rectMode()**: rect()の基準点の位置を変更する関数。引数は1つで4つの種類がある
 - ▶ **CORNER**
 - rect()の第1引数と第2引数が表す基準点を長方形の左上の角の点にする
 - **デフォルト**
 - ▶ **CORNERS**
 - rect()の第1引数と第2引数が表す基準点を長方形の一方の角の点にし，第3引数と第4引数が表す基準点を反対側の角の点にする
 - ▶ **CENTER**
 - rect()の第1引数と第2引数が表す基準点を，第3引数が幅，第4引数が高さの長方形の中心にする
 - ▶ **RADIUS**
 - rect()の第1引数と第2引数が表す基準点を，第3引数が半分の幅，第4引数が半分の高さの長方形の中心にする



ellipseMode()

- **ellipseMode()**: ellipse()の基準点の位置を変更する関数。引数は1つで4つの種類がある
 - ▶ **CORNER**
 - ellipse()の第1引数と第2引数が表す基準点を楕円の外接長方形(バウンディングボックス)の左上の角の点にする
 - ▶ **CORNERS**
 - ellipse()の第1引数と第2引数が表す基準点を楕円の外接長方形(バウンディングボックス)の一方の角の点にし、第3引数と第4引数が表す基準点を反対側の角の点にする
 - ▶ **CENTER**
 - ellipse()の第1引数と第2引数が表す基準点を、第3引数が幅、第4引数が高さの楕円の中心にする
 - **デフォルト**
 - ▶ **RADIUS**
 - ellipse()の第1引数と第2引数が表す基準点を、第3引数が半分の幅、第4引数が半分の高さの楕円の中心にする



arc(1)



arc()の基本的な使い方

- **arc()**: 弧(円周の一部)を描く関数

- ▶ 第1引数: 中心のx座標, 第2引数: 中心のy座標
- ▶ 第3引数: 幅, 第4引数: 高さ
- ▶ 第5引数: 弧の始まりの角度[rad], 第6引数: 弧の終わりの角度[rad]
 - 第5引数の値 \leq 第6引数の値

例

```
size(300, 200);  
arc(100, 50, 50, 50, 0, HALF_PI);  
arc(200, 50, 50, 50, 0, PI + HALF_PI);  
arc(100, 150, 50, 50, PI, TWO_PI + HALF_PI);  
arc(200, 150, 50, 50, QUARTER_PI, PI + QUARTER_PI);
```





arc(2)


arc()のモード

- Processingでは、同じ名前で引数の数が異なる関数が存在する
 - ▶ (例)arc()は7番目の引数を指定することで描き方を変えることができる
 - ▶ 7番目の引数: モード

- OPEN
- CHORD
- PIE

例

```
size(300, 200);  
arc(100, 50, 50, 50, QUARTER_PI, TWO_PI - QUARTER_PI);  
arc(200, 50, 50, 50, QUARTER_PI, TWO_PI - QUARTER_PI, OPEN);  
arc(100, 150, 50, 50, QUARTER_PI, TWO_PI - QUARTER_PI, CHORD);  
arc(200, 150, 50, 50, QUARTER_PI, TWO_PI - QUARTER_PI, PIE);
```





radians()

- **radians()**: 角度を「度」から「ラジアン」へ変換する関数


例

```
size(300, 200);  
arc(100, 50, 50, 50, 0, HALF_PI);  
arc(200, 50, 50, 50, 0, PI + HALF_PI);  
arc(100, 150, 50, 50, PI, TWO_PI + HALF_PI);  
arc(200, 150, 50, 50, QUARTER_PI, PI + QUARTER_PI);
```



例

```
size(300, 200);  
arc(100, 50, 50, 50, 0, radians(90));  
arc(200, 50, 50, 50, 0, radians(270));  
arc(100, 150, 50, 50, radians(180), radians(450));  
arc(200, 150, 50, 50, radians(45), radians(225));
```






setup()

- Processingは**setup()**の中のコードを**最初に一度**だけ実行する
- **setup()**を用いることで関数定義が可能になる

例

```
void setup() {  
  size(200, 200);  
  drawPackman(width / 2, height / 2,  
               width / 2, height / 2,  
               30, 330);  
}  
  
void drawPackman(float x, float y,  
                 float w, float h,  
                 float aStart, float aEnd) {  
  
  arc(x, y,  
       w, h,  
       radians(aStart), radians(aEnd),  
       PIE);  
}
```

setup()の中でsize()を書く
場合、必ず**1行目**にすること!



Processingであらかじめ用意されている図
形を描画する関数の引数の型は基本的に**float**

多角形，曲線



sato@ise.aoyama.ac.jp

2019/4/9




beginShape(), endShape()

- 複数の点をつなぎ合わせて複雑な形を描く

- ① **beginShape();**
 - 新しい形を描き始める合図
- ② **vertex();**(複数回)
 - 図形の輪郭を表す点
 - 引数に座標を指定
- ③ **endShape();**
 - 描き終わりの合図

例

```
size(300, 150);  
beginShape();  
vertex(50, 75);  
vertex(100, 30);  
vertex(100, 50);  
vertex(250, 50);  
vertex(250, 100);  
vertex(100, 100);  
vertex(100, 120);  
vertex(50, 75);  
endShape(CLOSE);
```



endShape()の引数にCLOSEを指定すると、最初の点と最後の点が直線で結ばれ閉じた形になる



beginShape()のパラメータ

- beginShape()には，次のパラメータが指定可能

- ▶ POINTS

- 点を描く

- ▶ LINES

- 2点1組の線を描く

- ▶ TRIANGLES

- 3点1組の三角形を描く

- ▶ TRIANGLE_STRIP

- 1つずつずらしながら3点1組の三角形を描く

- ▶ TRIANGLE_FAN

- 1つずつずらしながら3点1組の三角形を描く
 - 最初の点は固定

- ▶ QUADS


- 4点1組の四角形を描く

- ▶ QUADS_STRIP

- 2つずつずらしながら4点1組の四角形を描く

例

```
size(300, 150);  
beginShape(TRIANGLE_FAN);  
vertex(50, 75);  
vertex(100, 30);  
vertex(100, 50);  
vertex(250, 50);  
vertex(250, 100);  
vertex(100, 100);  
vertex(100, 120);  
vertex(50, 75);  
endShape();
```





curve()(1)


- **curve()**: 曲線を描く関数

- ▶ 8つの引数で4つの点の座標を指定
- ▶ 指定された4点を結ぶ曲線の**中間の2点の間**の部分を描く

例

```
size(200, 200);  
curve(80, 50, 50, 150, 120, 180, 150, 150);
```

曲線の形状を定義する点を
制御点(コントロールポイント)という。curve()の制
御点は**最初の点**と**最後の点**



中間の2点を結ぶ部分以
外の曲線は描かれない!



curve()(2)



曲線全体を描く方法(1)

- 次のようにしてcurve()の曲線全体を描くことができる

例

```
size(200, 200);  
curve(80, 50, 80, 50, 50, 150, 120, 180); // ①  
curve(80, 50, 50, 150, 120, 180, 150, 150); // ②  
curve(50, 150, 120, 180, 150, 150, 150, 150); // ③
```

曲線を構成する点に沿ってcurve()を繰り返し使い徐々に曲線を描いていく。最初と最後の点(制御点)は2回続ける。これらを除く点は1つずつずらす





curveVertex()


曲線全体を描く方法(2)

- vertex()をcurveVertex()にすると点の間が曲線で結ばれる

例

```
size(200, 200);  
beginShape();  
curveVertex(80, 50);  
curveVertex(80, 50);  
curveVertex(50, 150);  
curveVertex(120, 180);  
curveVertex(150, 150);  
curveVertex(150, 150);  
endShape();
```

curve()と同様に
制御点は2回続ける






閉曲線

- 最後の3点を最初の3点と同じにすることで閉曲線が描ける

例

```
size(200, 200);  
beginShape();  
curveVertex(100, 20);  
curveVertex(120, 80);  
curveVertex(180, 100);  
curveVertex(120, 120);  
curveVertex(100, 180);  
curveVertex(80, 120);  
curveVertex(20, 100);  
curveVertex(80, 80);  
curveVertex(100, 20);  
curveVertex(120, 80);  
curveVertex(180, 100);  
endShape();
```

最後に最初の
3点を描く





直線と曲線が混在する図形

- **vertex()とcurveVertex()を併用→直線と曲線が混在する図形が描ける**


例

```
boolean isCr1Pnt = true; // control point
float pX = 0, pY = 0; // pointX, pointY
size(250, 200);
float r = height / 3; // radius
beginShape();
vertex(width / 2, height / 2);
for (int i = 30; i < 330; i++) {
  float theta = radians(float(i));
  pX = width / 2 + r * cos(theta);
  pY = height / 2 + r * sin(theta);
  if (isCr1Pnt) {
    vertex(pX, pY);
    isCr1Pnt = false;
  }
  curveVertex(pX, pY);
}
curveVertex(pX, pY);
vertex(width / 2, height / 2);
endShape();
```

切り替え所の
点を2回続ける

注

- ▶ **直線→曲線の切り替え**
- **vertex(x, y);→curveVertex(x, y);**
- ▶ **曲線→直線の切り替え**
- **curveVertex(x, y);→curveVertex(x, y);**



補足



sato@ise.aoyama.ac.jp

2019/4/9




width, height

- Processingであらかじめ用意されている変数宣言なしで使える特殊な変数(システム変数)にwidthとheightがある
 - ▶ width: ウィンドウの幅
 - ▶ height: ウィンドウの高さ
 - いずれもint型
- size()が実行されると自動的にwidthとheightに値が設定される
 - ▶ size()を省略した場合, 自動的に「size(100, 100)」が実行される→widthとheightの値はそれぞれ100になる

例

```
size(200, 150);  
line(0, 0, width, height);  
line(width, 0, 0, height);  
ellipse(width / 2, height / 2, width / 2, height / 2);
```





算術関数

関数名	機能
<code>abs(a)</code>	aの絶対値を求める
<code>sqrt(a)</code>	aの平方根を求める
<code>sq(a)</code>	aの2乗を求める
<code>pow(a, b)</code>	aのb乗を求める
<code>exp(a)</code>	e(ネイピア数)のa乗を求める
<code>log(a)</code>	自然対数を求める
<code>round(a)</code>	小数点以下を四捨五入する
<code>floor(a)</code>	小数点以下を切捨てる
<code>ceil(a)</code>	小数点以下を切上げる
<code>min(a, b)</code>	最小値を求める
<code>max(a, b)</code>	最大値を求める
<code>constrain(a, min, max)</code>	aをminとmaxの間に制限する
<code>mag(x, y)</code>	原点から座標(x, y)までの距離を求める
<code>dist(x1, y1, x2, y2)</code>	2点間の距離を求める
<code>lerp(a, b, c)</code>	aとbの間の比cで指定される値を求める
<code>norm(a, b, c)</code>	aを範囲b-cから0-1の範囲へ変換する
<code>map(a, b, c, d, e)</code>	aを範囲b-cから別の範囲d-eへ変換する

minとmaxには配列
を渡すこともできる



三角関数

関数名	機能
$\sin(a)$	正弦(サイン)を求める
$\cos(a)$	余弦(コサイン)を求める
$\tan(a)$	正接(タンジェント)を求める
$\text{asin}(a)$	逆正弦(アークサイン)を求める
$\text{acos}(a)$	逆余弦(アークコサイン)を求める
$\text{atan}(a)$	逆正接(アークタンジェント)を求める
$\text{atan2}(\textcolor{red}{y}, \textcolor{red}{x})$	座標を指定して角度を求める
$\text{degrees}\textcolor{red}{s}(a)$	角度の単位を「ラジアン」から「度」へ変換する
$\text{radians}\textcolor{red}{s}(a)$	角度の単位を「度」から「ラジアン」へ変換する

引数の順番に
注意(yが先)

最後の「s」を忘れずに



型変換関数

関数名	機能
char()	文字(1文字)に変換
str()	文字列に変換
int()	整数に変換
float()	浮動小数点数に変換
byte()	1バイトのデータに変換
boolean()	0はfalse，その他はtrueに変換